TABULATE Tutorial

Alan Eaton and Ken Howell

SAS Institute

PROC TABULATE displays descriptive statistics arranged in hierarchical tables. This overview illustrates many features by using them in examples. The tables and the SAS program statements that produced them appear in figures throughout the text.

All examples use hypothetical survey data on New Orleans restaurants and entrees. The following SAS code creates the input data set and formats for TABULATE.

```
DATA;
   INPUT REST ENTREE SEX TASTE PRICE TIME;
CARDS;
1  1  0  3  8.31   77
2  1  0  5  5.55  100
3  2  1  7  6.35   68
.  .  .  .    .    .
.  .  .  .    .    .
.  .  .  .    .    .
;

PROC FORMAT;
   VALUE RESTF    1 = ANTOINES
                  2 = GALATOIRES
                  3 = ARNAUDS;
   VALUE SEXF     0 = MALE
                  1 = FEMALE;
   VALUE ENTREEF  1 = TROUT VERONIQUE
                  2 = SEAFOOD GUMBO
                  3 = SHRIMP CREOLE
                  4 = STUFFED FLOUNDER;
```

The variables contain the following information:

REST	the restaurant where the respondent ate
ENTREE	the entree eaten
SEX	sex of the respondent
TASTE	a measure of how much the respondent liked the entree
PRICE	the price of the entree
TIME	how long the respondent was in the restaurant.

We emphasize that the data is hypothetical and make no guarantees that the entrees mentioned are in fact served at the restaurants mentioned. Furthermore, prices and taste measurements were randomly generated and do not necessarily indicate the true price or taste quality of any entree.

TABULATE makes tables organized into classification hierarchies. Later we will show how to specify the items in a hierarchy. For now notice that the table in figure 1 displays the number of responses and the mean response value of TASTE. There are categories for male and female respondents for each entree at each restaurant. The table has three dimensions -- a page (only one for TASTE), rows, and columns. Four elements define each of these table cells:

- classification levels (of REST, SEX, and ENTREE)
- a single analysis variable (TASTE)
- a single statistic (N or MEAN)
- a format specification (described later).

The principal statements used with TABULATE are CLASS, VAR, and TABLE. Other important statements include KEYLABEL, LABEL, FORMAT, TITLE, FREQ, WEIGHT, and BY.

You can arrange the tables in many ways. At the highest level you can arrange them into page, row, and column dimensions. Figure 2 emphasizes first the column dimension, then the row dimension. A single class variable, REST, defines the classification hierarchy. Class variables must appear in a CLASS statement. These variables supply the levels that appear in table dimensions. REST produces three levels: ANTOINES, GALATOIRES, and ARNAUDS.

The TABLE statement indicates what classification levels to display in the table and where to put them. In the first table REST is alone in the column dimension. N is the number of observations in each sub-group defined by the class variable levels -- for 60

observations REST has the value ANTOINES. N is the default statistic when no analysis variable or other statistic applies.

The second TABLE statement places REST in the row dimension and explicitly calls for N in the column dimension. Separate dimension expressions with a comma. The values are the same as in the first table.

The TABLE statement of figure 3 locates REST in the page dimension, N in the row dimension, and ALL in the column dimension. There are separate wafers for each level of REST. Each wafer is a separate physical page. We will explain ALL below.

You can make tables more attractive for reports by replacing variable names and keywords with other text. Attach literal modifiers up to 40 characters long to any variable or keyword in a TABLE statement. For example, REST='RESTAURANT' in the TABLE statement of figure 4 replaces REST with RESTAURANT in the table. The KEYLABEL statement associates a label up to 40 characters long with statistic keywords or ALL. The KEYLABEL statement of figure 4 replaces N with COUNT.

Note that both invocations of PROC TABULATE in figure 4 produce the same table. Variable names can also be labelled with a label statement. The same effect can often be produced in a number of ways.

All statements other than TABLE and KEYLABEL statements must precede any KEYLABEL or TABLE statements. TABLE and KEYLABEL statements may occur in any order.

The next topics are table operations and grouping. Table operations are concatenation and nesting. Figure 5 shows a table similar to the restaurant frequency table. This table contains entree frequencies. Concatenation means that one table follows another. Concatenate two sub-tables by following the specification of one with the specification of the other: as in figure 6. The concatenation operator is also called the follow operator.

Figure 7 introduces a third concatenation element, the universal classifier, ALL. The universal classifier collapses all class levels into one level at the hierarchy level where it occurs. In figure 7 there is only one hierarchy level in any sub-table, either REST or ENTREE. ALL collapses the class levels into one level and shows that there are 303 respondents.

60 + 99 + 104 = 303
76 + 72 + 72 + 83 = 303.

The table in figure 8 follows REST with ENTREE in the row dimension. SEX is in the column dimension. Now concatenate ALL in each dimension (figure 9). ALL in the column dimension collapses the levels of SEX: 60=35+25. ALL in the row dimension collapses the levels of REST and ENTREE: 35+49+73=157, 38+37+38+44=157. The ALL,ALL table cell collapses the levels of both dimensions and contains the count of all respondents in the table.

The nesting operator, *, arranges all levels of one class within each level of another class to produce statistics on sub-groups of the data. In figure 10 both levels of SEX are nested within each level of REST. 35 males and 25 females ate at Antoines. The same counts appeared in figure 8 but arranged in two dimensions. In terms of establishing class interactions, the column dimension is nested within the row dimension and the row dimension is nested within the page dimension.

You can bind the nesting operator to a concatenation by grouping the concatenation within parentheses. Figure 11 shows REST and ENTREE both nested above SEX. Figure 12 shows what happens in the absence of grouping. You can concatenate and group nestings and there can be many levels of grouping.

The VAR statement specifies a list of analysis variables. Analysis variables supply values on which statistics are computed. When analysis variables are in a table statement the default statistic is SUM, the sum of the analysis variable values within a classification sub-group. Figure 13 shows the total amount of money and time spent on eating each entree in each restaurant as well as the total amount of time spent in each restaurant.

Include analysis variables in a table by nesting and concatenating them as

though they were class variables. Within a table all analysis variables must be grouped in the same dimension.

You can request the same statistics from TABULATE that you can request from PROC SUMMARY.

N	USS
NMISS	CSS
MEAN	STDERR
STD	CV
MIN	T
MAX	PRT
RANGE	VAR
SUM	SUMWGT

Request statistics by nesting statistic keywords in the TABLE statement. You must group all statistics in one table dimension. Statistics may appear in the same dimension as analysis variables or in a different dimension.

You can request two types of percents: PCTN and PCTSUM. PCTN is percent of N, the number of observations in a group. The percentages calculated are for each of the classification levels that contribute to a group total. With no further specification the group total is the total for all observations. In figure 14 the percentages are for each level of REST. The group total is the combined frequencies of all levels of REST.

PCTSUM is the percentage that a particular SUM value is of the of the total SUM value for an analysis variable in a sub-group. The following examples use PCTN but the remarks apply to both types of percents.

The next example (figure 15) is more complicated but the idea is the same. The denominator is the total of all N table cells -- 303. Percent cells are the percent that each N cell is of 303.

When you specify a denominator base you indicate which class levels to collapse into a denominator total. To include all values in a row in a total, as in figure 16, collapse all column class levels by making the entire column nesting the denominator base. The denominator base is SEX. For each level of REST collapse the two levels of SEX into a denominator total. 35 is 58.33% of 60.

Include all values in a column in the denominator by making the row nesting the denominator base. In figure 17 the denominator base is REST. For males the denominator is 157. 35 is 22.29% of 157.

Incorporate denominator values into the table by using ALL to collapse levels within a nesting. ALL in the column dimension of figure 18 collapses the levels of SEX within levels of REST to give row totals. Notice that two separate nestings define sub-sections of this table: REST*SEX and REST*ALL. Since PCTN applies to both sub-sections, both nestings must be represented among the denominator bases.

ALL in the row dimension (figure 19) collapses levels of SEX to give column totals. The table shows denominators for percents of column totals.

Figure 20 shows percents within independent sub-tables. No denominator is specified, so the denominator in each case is the total of N cells in the sub-table. In this table all denominators are 303.

The next three tables illustrate how nestings behave as denominator bases. PCTN is in the page dimension. In figure 21 the entire row nesting is the denominator base. This base collapses all levels of REST*ENTREE within each level of SEX into denominator totals. Each table cell is the percentage of the column total. Only percentages appear in the table. You do not have to ask for N when you ask for PCTN. Each column represents 100%.

Before continuing notice the effect of blank literal modifiers on the row and column titles. You can eliminate unwanted names and sometimes compress the page space required for a table by associating a name with a blank literal.

When you specify only part of a nesting in the denominator the related percents can become scattered in the table. Figure 22 shows the effect of specifying only REST. This base collapses levels of REST within each level of ENTREE*SEX into denominator totals. For example, one denominator total is the collapse of all REST levels for males who ate Trout Veronique. 21.05% of male respondents who ate Trout Veronique ate it at Antoines, 26.32% at Galatoires, and 52.63% at Arnauds.

The relationship is easier to see in figure 23 where the order of the row nesting is reversed. In general, if the denominator base covers classes in a nesting closest to an edge of the table related percentages are more closely grouped in the table.

We now consider formatting the numeric table cell values. You have seen the effect of formats throughout the examples. Use F=w.d or FORMAT=w.d to specify a format. Only w.d formats are currently available. Soon all SAS formats (e.g., COMMA7, DOLLAR10) will be available in TABULATE. You will specify them in the same way: F=COMMA7. We call F=w.d a format item. Nest format items in TABLE statement expressions. The default format is 12.2. In figure 24 the default applies to the MEAN columns and F=9 applies to the N columns.

You can change the default for all tables in the PROC step by using FORMAT= or F= on the PROC statement. In figure 25, F=9 is the default and applies to the N columns. F=8.2 applies to the MEAN columns.

Formats in the column dimension override formats in the row dimension; formats in the row dimension override formats in the page dimension. F=20 is the default format in figure 26 but the page dimension format always overrides it. Likewise, F=12.8 in the page dimension is always overridden by row and column formats. F=8.2 in the row dimension applies to the MEAN columns. F=9 in the column dimension applies to the N columns.

Column widths depend on the format widths. The width of the column is the width of the widest format that applies to the column. In figure 27, F=5 applies to the top nesting and F=7.2 applies to the bottom nesting. The width of the column is 7. The number of decimal places is always the number specified in the format.

Next we look at formatting the titles. Recall the PROC FORMAT program statements shown earlier. Use PROC FORMAT and FORMAT statements to format class variable values. TABULATE determines how much space is available to arrange formatted values over appropriate columns or beside appropriate rows, and automatically breaks character strings into as many lines as are needed.

The amount of space allotted to row titles is 1/4 of the linesize unless you specify an RTS value as a TABLE statement option. TABULATE allocates the same amount of space to each of the pieces in the row title. In figure 28 there are two pieces. RTS is 23 to allow 10 spaces for each piece and 3 spaces for vertical separation characters. Row title borders are included in row title space.

Use FORMCHAR= on the PROC statement to alter boxing characters in the table. Specify any 11 character string either as normal characters or as a hexadecimal string. Most of our examples use the IBM6670 boxing characters recommended in our documentation. The default string uses vertical bars, hyphens, and the plus sign. Figure 29 uses a blank FORMCHAR string. You can emphasize horizontal associations by replacing the plus sign corner character with a hyphen (figure 30).

Specify the order in which class levels appear in the table with the ORDER= option on the PROC statement. The possible orders are

FREQ	by descending frequency (N)
DATA	by the order in the data set
INTERNAL	by the internal representation of the data
FORMATTED	by the formatted (external) representation of the data.

In figure 31, ORDER=INTERNAL. Examine the PROC FORMAT statements to determine the unformatted values. In figure 32, ORDER=FORMATTED. ORDER=INTERNAL is the default order.

A missing class variable value normally causes the observation to be ignored. You can include missing class values as classification levels by specifying MISSING on the PROC statement.

```
PROC TABULATE MISSING;
```

Missing table cells result from missing analysis variable values for all interactions in a sub-group. You can substitute up to 20 characters of text for the missing value by using MISSTEXT= on the TABLE statement. The text is truncated to the column width.

If a row contains only missing values it will not print unless you specify PRINTMISS on the TABLE statement. Figure 33 illustrates both MISSTEXT and PRINTMISS. We arbitrarily set all TASTE values for Galatoires to missing. If PRINTMISS had not been specified the GALATOIRES row would not have printed. A wafer that consists entireiy of missing values will never print.

QUALITY OF TASTE

		SEX OF TASTER			
		MALE		FEMALE	
RESTAURANT	ENTREE	RESPONSES	MEAN RESPONSE	RESPONSES	MEAN RESPONSE
ANTOINES	TROUT VERONIQUE	8	5.13	4	4.75
	SEAFOOD GUMBO	7	6.43	6	7.17
	SHRIMP CREOLE	11	6.27	7	5.14
	STUFFED FLOUNDER	9	4.44	8	4.50
GALATOIRES	TROUT VERONIQUE	10	5.90	13	4.92
	SEAFOOD GUMBO	12	4.33	9	6.44
	SHRIMP CREOLE	11	5.55	13	5.15
	STUFFED FLOUNDER	16	6.00	15	5.13
ARNAUDS	TROUT VERONIQUE	20	5.10	21	4.71
	SEAFOOD GUMBO	18	4.56	20	6.55
	SHRIMP CREOLE	16	4.44	14	5.71
	STUFFED FLOUNDER	19	5.21	16	6.31

```
PROC TABULATE;
   CLASS REST SEX ENTREE;
   VAR TASTE;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   FORMAT SEX SEXF.;
   TABLE TASTE, REST*ENTREE,
      SEX*(N='RESPONSES'*F=9 MEAN='MEAN RESPONSE'*F=8.2)
      / RTS=23;
```

FIGURE 1

REST		
ANTOINES	GALATOIRES	ARNAUDS
N	N	N
60	99	144

	N
REST	
ANTOINES	60
GALATOIRES	99
ARNAUDS	144

```
PROC TABULATE F=10;
   FORMAT REST RESTF.;
   CLASS REST;
   TABLE REST;
   TABLE REST, N;
```

FIGURE 2

REST ANTOINES

	ALL
N	60

REST GALATOIRES

	ALL
N	99

REST ARNAUDS

	ALL
N	144

```
PROC TABULATE F=10;
   FORMAT REST RESTF.;
   CLASS REST;
   TABLE REST, N, ALL;
```

FIGURE 3

RESTAURANT		
ANTOINES	GALATOIRES	ARNAUDS
COUNT	COUNT	COUNT
60	99	144

```
PROC TABULATE F=10;
   FORMAT REST RESTF.;
   CLASS REST;
   TABLE REST='RESTAURANT'*N='COUNT';
```

```
PROC TABULATE F=10;
   FORMAT REST RESTF.;
   CLASS REST;
   LABEL REST = RESTAURANT;
   KEYLABEL N = COUNT;
   TABLE REST;
```

FIGURE 4

ENTREE			
TROUT VERONIQUE	SEAFOOD GUMBO	SHRIMP CREOLE	STUFFED FLOUNDER
COUNT	COUNT	COUNT	COUNT
76	72	72	83

```
PROC TABULATE F=10;
   FORMAT ENTREE ENTREEF.;
   CLASS ENTREE;
   KEYLABEL N = COUNT;
   TABLE ENTREE;
```

FIGURE 5

RESTAURANT			ENTREE			
ANTOINES	GALATOIRES	ARNAUDS	TROUT VERONIQUE	SEAFOOD GUMBO	SHRIMP CREOLE	STUFFED FLOUNDER
COUNT	COUNT	COUNT	COUNT	COUNT	COUNT	COUNT
60	99	144	76	72	72	83

```
PROC TABULATE F=10;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   CLASS REST ENTREE;
   KEYLABEL N = COUNT;
   TABLE REST='RESTAURANT' ENTREE*F=9;
```

FIGURE 6

RESTAURANT			ENTREE				
ANTOINES	GALATOIRES	ARNAUDS	TROUT VERONIQUE	SEAFOOD GUMBO	SHRIMP CREOLE	STUFFED FLOUNDER	ALL
COUNT	COUNT	COUNT	COUNT	COUNT	COUNT	COUNT	COUNT
60	99	144	76	72	72	83	303

```
PROC TABULATE F=10;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   CLASS REST ENTREE;
   KEYLABEL N = COUNT;
   TABLE REST='RESTAURANT' ENTREE*F=9 ALL*F=5;
```

FIGURE 7

	SEX	
	MALE	FEMALE
	COUNT	COUNT
RESTAURANT		
ANTOINES	35	25
GALATOIRES	49	50
ARNAUDS	73	71
ENTREE		
TROUT VERONIQUE	38	38
SEAFOOD GUMBO	37	35
SHRIMP CREOLE	38	34
STUFFED FLOUNDER	44	39

```
PROC TABULATE F=10;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   FORMAT SEX SEXF.;
   CLASS REST ENTREE SEX;
   KEYLABEL N = COUNT;
   TABLE REST='RESTAURANT' ENTREE,
      SEX / RTS=12;
```

FIGURE 8

	ALL	SEX	
		MALE	FEMALE
	COUNT	COUNT	COUNT
ALL	303	157	146
RESTAURANT			
ANTOINES	60	35	25
GALATOIRES	99	49	50
ARNAUDS	144	73	71
ENTREE			
TROUT VERONIQUE	76	38	38
SEAFOOD GUMBO	72	37	35
SHRIMP CREOLE	72	38	34
STUFFED FLOUNDER	83	44	39

```
PROC TABULATE F=10;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   FORMAT SEX SEXF.;
   CLASS REST ENTREE SEX;
   KEYLABEL N = COUNT;
   TABLE ALL REST='RESTAURANT' ENTREE,
      ALL SEX / RTS=12;
```

FIGURE 9

RESTAURANT					
ANTOINES		GALATOIRES		ARNAUDS	
SEX		SEX		SEX	
MALE	FEMALE	MALE	FEMALE	MALE	FEMALE
COUNT	COUNT	COUNT	COUNT	COUNT	COUNT
35	25	49	50	73	71

```
PROC TABULATE F=10;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   FORMAT SEX SEXF.;
   CLASS REST ENTREE SEX;
   KEYLABEL N = 'COUNT';
   TABLE REST='RESTAURANT'*SEX;
```

FIGURE 10

		COUNT
RESTAURANT	SEX	
ANTOINES	MALE	35
	FEMALE	25
GALATOIRES	MALE	49
	FEMALE	50
ARNAUDS	MALE	73
	FEMALE	71
ENTREE	SEX	
TROUT VERONIQUE	MALE	38
	FEMALE	38
SEAFOOD GUMBO	MALE	37
	FEMALE	35
SHRIMP CREOLE	MALE	38
	FEMALE	34
STUFFED FLOUNDER	MALE	44
	FEMALE	39

```
PROC TABULATE F=10;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   FORMAT SEX SEXF.;
   CLASS REST ENTREE SEX;
   KEYLABEL N = 'COUNT';
   TABLE (REST='RESTAURANT' ENTREE)*SEX,
      N*F=6 / RTS=23;
```

FIGURE 11

	COUNT
RESTAURANT	
ANTOINES	60
GALATOIRES	99
ARNAUDS	144
ENTREE	SEX
TROUT VERONIQUE	MALE 38
	FEMALE 38
SEAFOOD GUMBO	MALE 37
	FEMALE 35
SHRIMP CREOLE	MALE 38
	FEMALE 34
STUFFED FLOUNDER	MALE 44
	FEMALE 39

```
PROC TABULATE F=10;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   FORMAT SEX SEXF.;
   CLASS REST ENTREE SEX;
   KEYLABEL N = 'COUNT';
   TABLE REST='RESTAURANT' ENTREE*SEX,
      N*F=6 / RTS=23;
```

FIGURE 12

	ENTREE								
	TROUT VERONIQUE		SEAFOOD GUMBO		SHRIMP CREOLE		STUFFED FLOUNDER		TIME
	PRICE	TIME	PRICE	TIME	PRICE	TIME	PRICE	TIME	
	SUM	SUM	SUM	SUM	SUM	SUM	SUM	SUM	SUM
RESTAURANT									
ANTOINES	82.80	987	99.82	1141	38.84	1642	73.65	1595	5365
GALATOIRES	132.84	1923	132.93	1940	104.08	2096	172.62	2706	8665
ARNAUDS	204.76	3761	218.20	3356	153.45	2589	240.34	3041	12747

```
PROC TABULATE F=6;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   FORMAT SEX SEXF.;
   CLASS REST ENTREE SEX;
   VAR PRICE TIME;
   TABLE REST='RESTAURANT', ENTREE*(PRICE*F=6.2 TIME) TIME
      / RTS=12;
```

FIGURE 13

RESTAURANT					
ANTOINES		GALATOIRES		ARNAUDS	
N	PCTN	N	PCTN	N	PCTN
60	19.80	99	32.67	144	47.52

60 + 99 + 144 = 303

```
PROC TABULATE F=6;
   CLASS REST;
   LABEL REST = 'RESTAURANT';
   FORMAT REST RESTF.;
   TABLE REST*(N PCTN*F=6.2);
```

FIGURE 14

	SEX OF TASTER			
	MALE		FEMALE	
	N	PCTN	N	PCTN
RESTAURANT				
ANTOINES	35	11.55	25	8.25
GALATOIRES	49	16.17	50	16.50
ARNAUDS	73	24.09	71	23.43

35 + 49 + 73 + 25 + 50 + 71 = 303

```
PROC TABULATE F=6;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   CLASS REST SEX;
   TABLE REST='RESTAURANT',
      SEX='SEX OF TASTER'*(N PCTN*F=6.2)
      / RTS=12;
```

FIGURE 15

	SEX OF TASTER			
	MALE		FEMALE	
	N	PCTN	N	PCTN
RESTAURANT				
ANTOINES	35	58.33	25	41.67
GALATOIRES	49	49.49	50	50.51
ARNAUDS	73	50.69	71	49.31

35 + 25 = 60

49 + 50 = 99

73 + 71 = 144

```
PROC TABULATE F=6;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   CLASS REST SEX;
   TABLE REST='RESTAURANT',
      SEX='SEX OF TASTER'*
      (N PCTN<SEX>*F=6.2) / RTS=12;
```

FIGURE 16

	SEX OF TASTER			
	MALE		FEMALE	
	N	PCTN	N	PCTN
RESTAURANT				
ANTOINES	35	22.29	25	17.12
GALATOIRES	49	31.21	50	34.25
ARNAUDS	73	46.50	71	48.63

35 + 49 + 73 = 157

25 + 50 + 71 = 146

```
PROC TABULATE F=6;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   CLASS REST SEX;
   TABLE REST='RESTAURANT',
      SEX='SEX OF TASTER'*
      (N PCTN<REST>*F=6.2) / RTS=12;
```

FIGURE 17

	SEX OF TASTER					
	MALE		FEMALE		BOTH	
	N	PCTN	N	PCTN	N	PCTN
RESTAURANT						
ANTOINES	35	58.33	25	41.67	60	100.00
GALATOIRES	49	49.49	50	50.51	99	100.00
ARNAUDS	73	50.69	71	49.31	144	100.00

35 + 25 = 60

49 + 50 = 99

73 + 71 = 144

```
PROC TABULATE F=6;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   CLASS REST SEX;
   TABLE REST='RESTAURANT',
      (SEX='SEX OF TASTER' ALL='BOTH')*
      (N PCTN<SEX ALL>*F=6.2) / RTS=12;
```

FIGURE 18

SAS

	SEX OF TASTER			
	MALE		FEMALE	
	N	PCTN	N	PCTN
RESTAURANT				
ANTOINES	35	22.29	25	17.12
GALATOIRES	49	31.21	50	34.25
ARNAUDS	73	46.50	71	48.63
ALL	157	100.00	146	100.00

35 + 49 + 73 = 157

25 + 50 + 71 = 146

```
PROC TABULATE F=6;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   CLASS REST SEX;
   TABLE REST='RESTAURANT' ALL,
      SEX='SEX OF TASTER'
      (N PCTN<REST ALL>*F=6.2) / RTS=12;
```

FIGURE 19

	SEX OF TASTER					
	MALE		FEMALE		BOTH	
	N	PCTN	N	PCTN	N	PCTN
RESTAURANT						
ANTOINES	35	11.55	25	8.25	60	19.80
GALATOIRES	49	16.17	50	16.50	99	32.67
ARNAUDS	73	24.09	71	23.43	144	47.52
ALL	157	51.82	146	48.18	303	100.00

35 + 49 + 73 + 25 + 50 + 71 = 303

60 + 99 + 144 = 303

157 + 146 = 303

```
PROC TABULATE F=6;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   CLASS REST SEX;
   TABLE REST='RESTAURANT' ALL,
      (SEX='SEX OF TASTER' ALL='BOTH')*
      (N PCTN*F=6.2) / RTS=12;
```

FIGURE 20

PCTN

		MALE	FEMALE
ANTOINES	TROUT VERONIQUE	5.10	2.74
	SEAFOOD GUMBO	4.46	4.11
	SHRIMP CREOLE	7.01	4.79
	STUFFED FLOUNDER	5.73	5.48
GALATOIRES	TROUT VERONIQUE	6.37	8.90
	SEAFOOD GUMBO	7.64	6.16
	SHRIMP CREOLE	7.01	8.90
	STUFFED FLOUNDER	10.19	10.27
ARNAUDS	TROUT VERONIQUE	12.74	14.38
	SEAFOOD GUMBO	11.46	13.70
	SHRIMP CREOLE	10.19	9.59
	STUFFED FLOUNDER	12.10	10.96

```
PROC TABULATE;
   CLASS REST SEX ENTREE;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   FORMAT SEX SEXF.;
   TABLE PCTN<REST*SEX>*F=6.2,
         REST=' '*ENTREE=' ', SEX=' ' / RTS=35;
```

FIGURE 21

PCTN

		MALE	FEMALE
ANTOINES	TROUT VERONIQUE	21.05	10.53
	SEAFOOD GUMBO	18.92	17.14
	SHRIMP CREOLE	28.95	20.59
	STUFFED FLOUNDER	20.45	20.51
GALATOIRES	TROUT VERONIQUE	26.32	34.21
	SEAFOOD GUMBO	32.43	25.71
	SHRIMP CREOLE	28.95	38.24
	STUFFED FLOUNDER	36.36	38.46
ARNAUDS	TROUT VERONIQUE	52.63	55.26
	SEAFOOD GUMBO	48.65	57.14
	SHRIMP CREOLE	42.11	41.18
	STUFFED FLOUNDER	43.18	41.03

```
PROC TABULATE;
   CLASS REST SEX ENTREE;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   FORMAT SEX SEXF.;
   TABLE PCTN<REST>*F=6.2,
         REST=' '*ENTREE=' ', SEX=' ' / RTS=35;
```

FIGURE 22

PCTN

		MALE	FEMALE
TROUT VERONIQUE	ANTOINES	21.05	10.53
	GALATOIRES	26.32	34.21
	ARNAUDS	52.63	55.26
SEAFOOD GUMBO	ANTOINES	18.92	17.14
	GALATOIRES	32.43	25.71
	ARNAUDS	48.65	57.14
SHRIMP CREOLE	ANTOINES	28.95	20.59
	GALATOIRES	28.95	38.24
	ARNAUDS	42.11	41.18
STUFFED FLOUNDER	ANTOINES	20.45	20.51
	GALATOIRES	36.36	38.46
	ARNAUDS	43.18	41.03

```
PROC TABULATE;
   CLASS REST SEX ENTREE;
   FORMAT REST RESTF.;
   FORMAT ENTREE ENTREEF.;
   FORMAT SEX SEXF.;
   TABLE PCTN<REST>*F=6.2,
         ENTREE=' '*REST=' ', SEX=' ' / RTS=35;
```

FIGURE 23

TASTE

	SEX			
	MALE		FEMALE	
	RESPONSES	MEAN RESPONSE	RESPONSES	MEAN RESPONSE
RESTAURANT				
ANTOINES	35	5.57	25	5.36
GALATOIRES	49	5.47	50	5.32
ARNAUDS	73	4.85	71	5.79

```
PROC TABULATE;
   CLASS REST SEX;
   VAR TASTE;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   LABEL REST = 'RESTAURANT';
   TABLE TASTE, REST,
         SEX*(N='RESPONSES'*F=9 MEAN='MEAN RESPONSE')
         / RTS=15;
```

FIGURE 24

TASTE

	SEX			
	MALE		FEMALE	
	RESPONSES	MEAN RESPONSE	RESPONSES	MEAN RESPONSE
RESTAURANT				
ANTOINES	35	5.57	25	5.36
GALATOIRES	49	5.47	50	5.32
ARNAUDS	73	4.85	71	5.79

```
PROC TABULATE F=9;
   CLASS REST SEX ENTREE;
   VAR TASTE PRICE;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   FORMAT ENTREE ENTREEF.;
   LABEL REST = 'RESTAURANT';
   TABLE TASTE, REST,
         SEX*(N='RESPONSES' MEAN='MEAN RESPONSE'*F=8.2)
         / RTS=15;
```

FIGURE 25

TASTE

	SEX			
	MALE		FEMALE	
	RESPONSES	MEAN RESPONSE	RESPONSES	MEAN RESPONSE
RESTAURANT				
ANTOINES	35	5.57	25	5.36
GALATOIRES	49	5.47	50	5.32
ARNAUDS	73	4.85	71	5.79

```
PROC TABULATE F=20;
   CLASS REST SEX ENTREE;
   VAR TASTE PRICE;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   FORMAT ENTREE ENTREEF.;
   LABEL REST = 'RESTAURANT';
   TABLE TASTE*F=12.8, REST*F=8.2,
         SEX*(N='RESPONSES'*F=9 MEAN='MEAN RESPONSE')
         / RTS=15;
```

FIGURE 26

		MEAN
RESTAURANT		
ANTOINES	TASTE	5
GALATOIRES	TASTE	5
ARNAUDS	TASTE	5
ENTREE		
TROUT VERONIQUE	PRICE	5.53
SEAFOOD GUMBO	PRICE	6.26
SHRIMP CREOLE	PRICE	4.12
STUFFED FLOUNDER	PRICE	5.86

```
PROC TABULATE F=9;
   CLASS REST SEX ENTREE;
   VAR TASTE PRICE;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   FORMAT ENTREE ENTREEF.;
   LABEL REST = 'RESTAURANT';
   TABLE REST*TASTE*F=5 ENTREE*PRICE*F=7.2,
         MEAN / RTS=25;
```

FIGURE 27

TASTE

		MEAN
ANTOINES	MALE	5.57
	FEMALE	5.36
GALATOIRES	MALE	5.47
	FEMALE	5.32
ARNAUDS	MALE	4.85
	FEMALE	5.79
TROUT VERONIQUE	MALE	5.32
	FEMALE	4.79
SEAFOOD GUMBO	MALE	4.84
	FEMALE	6.63
SHRIMP CREOLE	MALE	5.29
	FEMALE	5.38
STUFFED FLOUNDER	MALE	5.34
	FEMALE	5.49

```
PROC TABULATE;
   CLASS REST SEX ENTREE;
   VAR TASTE;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   FORMAT ENTREE ENTREEF.;
   TABLE TASTE, (REST=' ' ENTREE=' ')*SEX=' ',
         MEAN*F=9.2 / RTS=23;
```

FIGURE 28

TASTE

		MEAN
ANTOINES	MALE	5.57
	FEMALE	5.36
GALATOIRES	MALE	5.47
	FEMALE	5.32
ARNAUDS	MALE	4.85
	FEMALE	5.79
TROUT VERONIQUE	MALE	5.32
	FEMALE	4.79
SEAFOOD GUMBO	MALE	4.84
	FEMALE	6.63
SHRIMP CREOLE	MALE	5.29
	FEMALE	5.38
STUFFED FLOUNDER	MALE	5.34
	FEMALE	5.49

```
PROC TABULATE FORMCHAR=' ';
   CLASS REST SEX ENTREE;
   VAR TASTE;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   FORMAT ENTREE ENTREEF.;
   TABLE TASTE, (REST=' ' ENTREE=' ')*SEX=' ',
         MEAN*F=9.2 / RTS=23;
```

FIGURE 29

MEAN OF TASTE

ANTOINES	MALE	5.57
	FEMALE	5.36
GALATOIRES	MALE	5.47
	FEMALE	5.32
ARNAUDS	MALE	4.85
	FEMALE	5.79
TROUT VERONIQUE	MALE	5.32
	FEMALE	4.79
SEAFOOD GUMBO	MALE	4.84
	FEMALE	6.63
SHRIMP CREOLE	MALE	5.29
	FEMALE	5.38
STUFFED FLOUNDER	MALE	5.34
	FEMALE	5.49

```
PROC TABULATE FORMCHAR='|----|-|---';
   CLASS REST SEX ENTREE;
   VAR TASTE;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   FORMAT ENTREE ENTRFEF.;
   TABLE TASTE*MEAN,
         (REST=' ' ENTREE=' ')*SEX=' ',
         ALL*F=9.2 / RTS=23;
```

FIGURE 30

TASTE

		MEAN
ANTOINES	MALE	5.57
	FEMALE	5.36
GALATOIRES	MALE	5.47
	FEMALE	5.32
ARNAUDS	MALE	4.85
	FEMALE	5.79
TROUT VERONIQUE	MALE	5.32
	FEMALE	4.79
SEAFOOD GUMBO	MALE	4.84
	FEMALE	6.63
SHRIMP CREOLE	MALE	5.29
	FEMALE	5.38
STUFFED FLOUNDER	MALE	5.34
	FEMALE	5.49

```
PROC TABULATE ORDER=INTERNAL;
   CLASS REST SEX ENTREE;
   VAR TASTE;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   FORMAT ENTREE ENTREEF.;
   TABLE TASTE,
      (REST=' ' ENTREE=' ')*SEX=' ',
      MEAN*F=9.2 / RTS=23;
```

FIGURE 31

TASTE

		MEAN
ANTOINES	FEMALE	5.36
	MALE	5.57
ARNAUDS	FEMALE	5.79
	MALE	4.85
GALATOIRES	FEMALE	5.32
	MALE	5.47
SEAFOOD GUMBO	FEMALE	6.63
	MALE	4.84
SHRIMP CREOLE	FEMALE	5.38
	MALE	5.29
STUFFED FLOUNDER	FEMALE	5.49
	MALE	5.34
TROUT VERONIQUE	FEMALE	4.79
	MALE	5.32

```
PROC TABULATE ORDER=FORMATTED;
   CLASS REST SEX ENTREE;
   VAR TASTE;
   FORMAT REST RESTF.;
   FORMAT SEX SEXF.;
   FORMAT ENTREE ENTREEF.;
   TABLE TASTE,
      (REST=' ' ENTREE=' ')*SEX=' ',
      MEAN*F=9.2 / RTS=23;
```

FIGURE 32

TASTE

	MEAN RESPONSE
RESTAURANT	
ANTOINES	5.48
GALATOIRES	NO RESP
ARNAUDS	5.31

```
DATA B;
 SET A;
 IF REST=2 THEN TASTE=.;

PROC TABULATE DATA=B;
 CLASS REST;
 VAR TASTE;
 FORMAT REST RESTF.;
 TABLE TASTE,
    REST='RESTAURANT',
    MEAN='MEAN RESPONSE'*F=8.2
    / RTS=12 MISSTEXT='NO RESP'
    PRINTMISS;
```

FIGURE 33